



=====Arbitrary_waveform_generation=====

HOW TO GENERATE ANY INPUT WAVEFORM FOR TRANSIENT SIMULATION WITHIN A SINGLE SPICE FILE.



```
Circuit: Arbitrary_waveform_generation
=====Want_100_lus_steps=====
Total_Period_s = 0.0001
Bin_Resolutio_Hz = 10000
Sample_Period_s = 1E-06
Nyquist_Hz = 500000
=====Create_PWL_array_and_Index_and_Plot=====
=====Add_1Vrms_Noise_to_PWL_array=====
=====Install_the_PWL_array=====
=====Run_and_Plot=====
=====Find_Ave_Rms=====
Average level 0.110889
RMS level 1.10046
=====FFT_and_Plot=====
FFT_BandWidth_Hz= 1E+06
FFT_resolution_Hz= 10000
=====done=====
```

Arbitrary_waveform_generation

```
*=====Need_A_voltage_Source_to_alter=====
v1 v1 0 0 dc

.control
set pensize = 2
echo "=====Want_100_lus_steps=====
let n = 100
let tstep = lus
let period_t = n*tstep
let Bin_Hz = 1/period_t
let nyquist = .5/tstep
echo "Total_Period_s = $&period_t"
echo "Bin_Resolutio_Hz = $&Bin_Hz"
echo "Sample_Period_s = $&tstep"
echo "Nyquist_Hz = $&nyquist"
echo "=====Create_PWL_array_and_Index_and_Plot=====
let pwl_1 = vector(2*n)*tstep*0.5
let ii = vector(2*$&n)
*plot pwl_1 vs ii
echo "=====Add_1Vrms_Noise_to_PWL_array=====
let index = 0
repeat $&n
let pwl_1[1+2*index] = 1.2*(rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)+rnd(127)-507.5)/102.879
let index = index + 1
end
echo "=====Install_the_PWL_array=====
alter @v1[pwl] = pwl_1
echo "=====Run_and_Plot=====
let period_s = tstep/2
*tran 0.5us 100us
tran $&period_s $&period_t 0 $&period_s
plot v1
*plot v(V1) pointplot
echo "=====Find_Ave_Rms=====
let averVal = mean(v1)
let noisAC = v1 - averVal
let RmsVal = sqrt(mean(noisAC* noisAC))
echo "Average level $&averVal"
echo "RMS level $&RmsVal"
unlet averVal
unlet RmsVal
echo "=====FFT_and_Plot=====
linearize
let FFT_BandWidth_Hz = 1Meg
let FFT_resolution_Hz = 10k
echo "FFT_BandWidth_Hz= $&FFT_BandWidth_Hz"
echo "FFT_resolution_Hz= $&FFT_resolution_Hz"
set specwindow= "rectangular"
spec $&FFT_resolution_Hz $&FFT_BandWidth_Hz $&FFT_resolution_Hz v(v1)
let expect_V = (1/sqrt(500k/10k))/(1+(frequency/600k)*(frequency/600k)*(frequency/600k)*(frequency/600k)*(frequency/600k))
plot mag (v1) expect_V loglog
echo "=====done=====
.endc
.end
```